

پس از مدتی که از شروع به کار یک سیستم می‌گذرد، همانطور که تعریف ایندکس‌های مفید سرعت جستجوها را بالا می‌برد، ایجاد fragmentation در آن‌ها نیز تاثیر منفی در کارایی خواهد داشت. به همین منظور نیاز است هر از چندگاهی بررسی شود میزان fragmentation ایندکس‌ها چقدر است. اگر این میزان بیش از 30 درصد بود توصیه شده است که از دستور [DBCC INDEXDEFRAG](#) استفاده شود یا بازسازی مجدد ([rebuild](#)) ایندکس‌ها صورت گیرد.

یکی دیگر از امکانات dmV های اس کیوال سرورهای 2005 به بعد، ارائه آمار میزان fragmentation ایندکس‌ها است که کوئری آن به صورت زیر می‌تواند باشد:

```
USE dbName;
SELECT OBJECT_NAME(DMV.object_id) AS TABLE_NAME,
       SI.NAME AS INDEX_NAME,
       avg_fragmentation_in_percent AS FRAGMENT_PERCENT,
       DMV.record_count
FROM   sys.dm_db_index_physical_stats(DB_ID(), NULL, NULL, NULL, 'SAMPLED') AS
       DMV
       LEFT OUTER JOIN SYS.INDEXES AS SI
         ON DMV.OBJECT_ID = SI.OBJECT_ID
        AND DMV.INDEX_ID = SI.INDEX_ID
WHERE  avg_fragmentation_in_percent > 10
       AND index_type_desc IN ('CLUSTERED INDEX', 'NONCLUSTERED INDEX')
       AND DMV.record_count >= 2000
ORDER BY
       TABLE_NAME DESC
```

باید در نظر داشت که اجرای این کوئری بر روی یک دیتابیس حجیم زمان‌بر بوده و احتمالاً عملکرد سیستم را تحت تاثیر قرار می‌دهد. بنابراین استفاده از آن در خارج از ساعات کاری باید مد نظر باشد. بازسازی ایندکس‌ها نیز به همین صورت است.

برای بازسازی تمامی ایندکس‌های یک دیتابیس مفروض می‌توان از [کوئری زیر](#) استفاده کرد:

```
DECLARE @TableName VARCHAR(255)
DECLARE @sql NVARCHAR(500)
DECLARE @fillfactor INT
SET @fillfactor = 80
DECLARE TableCursor CURSOR
FOR
    SELECT OBJECT_SCHEMA_NAME([object_id]) + '.' + NAME AS TableName
    FROM   sys.tables

OPEN TableCursor
FETCH NEXT FROM TableCursor INTO @TableName
WHILE @@FETCH_STATUS = 0
BEGIN
    SET @sql = 'ALTER INDEX ALL ON ' + @TableName +
              ' REBUILD WITH (FILLFACTOR = ' + CONVERT(VARCHAR(3), @fillfactor) + ' )'

    EXEC (@sql)
    FETCH NEXT FROM TableCursor INTO @TableName
END
CLOSE TableCursor
DEALLOCATE TableCursor
```