

عنوان: بررسی صحت پشتیبان‌های تهیه شده در SQL Server

نویسنده: وحید نصیری

تاریخ: ۱۸:۳۸:۰۰ ۱۳۸۸/۰۱/۱۸

آدرس: www.dotnettips.info

برچسب‌ها: SQL Server

اولین و اساسی‌ترین قدم در نگهداری یک سیستم مبتنی بر داده، تهیه پشتیبان‌های منظم و همچنین قابل اطمینان می‌باشد.

دستور T-SQL زیر بدون ری‌استور کردن یک فایل بک آپ اس کیوال سرور، سعی در تعیین اعتبار آن می‌کند:

```
RESTORE VERIFYONLY  
FROM DISK = 'C:\SQL_Backup\Test1'  
WITH FILE = 1,  
LOADHISTORY
```

این دستور

وضعیت کامل بودن پشتیبان و همچنین قابل خواندن بودن اطلاعات آن را بررسی می‌کند و در صورت سالم بودن بک آپ، پیغام زیر را نمایش خواهد داد:

```
The backup set on file 1 is valid.
```

عموماً عملیات تهیه پشتیبان در یک مکان مشخص در سرور صورت می‌گیرد (خصوصاً اگر یک job مختص به آن تعریف شده باشد که این کار را به صورت خودکار انجام دهد). بنابراین می‌توان عملیات اعتبار سنجی فوق را مکانیزه کرد. اسکریپت زیر مسیر آخرین بک آپ‌های گرفته شده در سرور را بر می‌گرداند:

```
SELECT DISTINCT physical_device_name  
FROM msdb.dbo.backupmediafamily  
ORDER BY  
physical_device_name
```

اکنون می‌توان مسیرهای فوق را در یک cursor جهت بررسی صحت تک تک موارد استفاده نمود:

```
DECLARE @path NVARCHAR(1000),  
        @msg NVARCHAR(MAX),  
        @NewLine CHAR(2),  
        @sql NVARCHAR(2000)  
  
SET @NewLine = CHAR(13) + CHAR(10)  
SET @msg = ''  
  
DECLARE DATABASES_CURSOR CURSOR  
FOR  
    SELECT DISTINCT physical_device_name  
    FROM msdb.dbo.backupmediafamily  
    ORDER BY  
        physical_device_name  
  
OPEN DATABASES_CURSOR  
  
FETCH NEXT FROM DATABASES_CURSOR INTO @path  
  
WHILE @@FETCH_STATUS = 0  
BEGIN  
    PRINT 'Verifying: ' + @path  
    SET @sql = 'RESTORE VERIFYONLY FROM DISK = ''' + @path  
        + ''' WITH FILE = 1, LOADHISTORY'  
  
    EXEC sp_executesql @sql  
    IF @@ERROR <> 0  
    BEGIN  
        SET @msg = @msg + 'Failed to verify: ' + @path + @NewLine  
    END  
  
    FETCH NEXT FROM DATABASES_CURSOR INTO @path  
END
```

```
CLOSE DATABASES_CURSOR
DEALLOCATE DATABASES_CURSOR

IF @msg <> ''
BEGIN
    PRINT @msg
    -- send email
    EXECUTE msdb.dbo.sp_send_dbmail
    @recipients = 'nasiri@site.net', -- Change This
    @copy_recipients = 'Administrator@site.net', -- Change This
    @Subject = 'backup verification info.',
    @Body = @msg
    ,@importance = 'High'
END
```

اسکرپت فوق بر روی تمامی مسیرهای ثبت شده موجود که در آن‌ها بیشتر پشتیبان تهیه شده است، دستور RESTORE VERIFYONLY را اجرا می‌کند و در آخر اگر پیغامی حاصل شد، یعنی مشکلی پدید آمده و ایمیلی را به اشخاص مورد نظر ارسال می‌کند.

می‌توان بر روی این اسکرپت یک job تهیه کرد که هر روز پس از تهیه بک آپ خودکار، کار بررسی صحت عملیات را نیز انجام دهد.